NAG Fortran Library Routine Document

F07JDF (DPTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07JDF (DPTTRF) computes the modified Cholesky factorization of a real n by n symmetric positive-definite tridiagonal matrix A.

2 Specification

```
SUBROUTINE F07JDF (N, D, E, INFO)
INTEGER             N, INFO
double precision    D(*), E(*)
```

The routine may be called by its LAPACK name ***dpttrf***.

3 Description

F07JDF (DPTTRF) factorizes the matrix A as

$$A = LDL^{\mathrm{T}},$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix with positive diagonal elements. The factorization may also be regarded as having the form $U^{\mathrm{T}}DU$, where U is a unit upper bidiagonal matrix.

4 References

None.

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: D$(*)$ – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the matrix A.

On exit: is overwritten by the n diagonal elements of the diagonal matrix D from the LDL^{T} factorization of A.

3: E$(*)$ – ***double precision*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n-1)$ subdiagonal elements of the matrix A.

On exit: is overwritten by the $(n-1)$ subdiagonal elements of the lower bidiagonal matrix L. E can also be regarded as containing the $(n-1)$ superdiagonal elements of the upper bidiagonal matrix U.

4: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i is not positive-definite. If $i <$ N the factorization could not be completed, while if $i =$ N, the factorization was completed, but D(N) ≤ 0.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = LDL^{\mathrm{T}},$$

where

$$\|E\|_\infty = O(\epsilon)\|A\|_\infty$$

and ϵ is the ***machine precision***.

Following the use of this routine, F07JEF (DPTTRS) can be used to solve systems of equations $AX = B$, and F07JGF (DPTCON) can be used to estimate the condition number of A.

8 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n.

The complex analogue of this routine is F07JRF (ZPTTRF).

9 Example

To factorize the symmetric positive-definite tridiagonal matrix A given by

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix}.$$

9.1 Program Text

```
*     F07JDF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*     .. Local Scalars ..
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), E(NMAX-1)
*     .. External Subroutines ..
      EXTERNAL         DPTTRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07JDF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
```

```
      IF (N.LE.NMAX) THEN
*
*        Read the lower bidiagonal part of the tridiagonal matrix A from
*        data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Factorize the tridiagonal matrix A
*
         CALL DPTTRF(N,D,E,INFO)
*
         IF (INFO.GT.0) THEN
            WRITE (NOUT,99999) 'The leading minor of order ', INFO,
     +        ' is not positive definite'
         END IF
*
*        Print details of the factorization
*
         WRITE (NOUT,*) 'Details of factorization'
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' The diagonal elements of D'
         WRITE (NOUT,99998) (D(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     ' Sub-diagonal elements of the Cholesky factor L'
         WRITE (NOUT,99998) (E(I),I=1,N-1)
*
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I3,A)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,5I9)
      END
```

9.2 Program Data

```
F07JDF Example Program Data
  5                          :Value of N
  4.0  10.0  29.0  25.0   5.0 :End of diagonal D
 -2.0  -6.0  15.0   8.0       :End of sub-diagonal E
```

9.3 Program Results

```
 F07JDF Example Program Results

 Details of factorization

  The diagonal elements of D
    4.0000   9.0000  25.0000  16.0000   1.0000

  Sub-diagonal elements of the Cholesky factor L
   -0.5000  -0.6667   0.6000   0.5000
```